· 专题研究 ·

样本量估计及其在 nQuery 软件上的实现
——回归分析(二)

南方医科大学 公共卫生与热带医学学院 生物统计学系　潘　蕾　周映雪　陈方尧　陈平雁△

5.1.1.5　增加协变量的多元线性回归

方法:Cohen (1988)[2]给出增加协变量后多元线性回归的样本量估计建立在自由度分别为 b 和 $n-a-b-1$,非中心参数为$\frac{n(R_{AB}^2-R_A^2)}{1-R_{AB}^2}$的非中心 F 分布基础上,其检验效能的计算公式为:

$$1-\beta=1-\mathrm{Prob}F\left(f_{1-\alpha,b,n-a-b-1},b,n-a-b-1,\frac{n(R_{AB}^2-R_A^2)}{1-R_{AB}^2}\right) \quad (5-8)$$

式中,R_A^2 为包含 A 个协变量的原回归模型的决定系数;R_{AB}^2为在原模型基础上增加 B 个协变量后的决定系数。

在计算样本量时,一般先设定样本量初始值,然后迭代样本量直到所得的检验效能满足条件为止。此时的样本量,即研究所需的样本量。

〔例 5-5〕某研究拟用多元线性回归模型分析地方医院的医务人员需求量,考虑变量如下:应变量 Y 为医务人员需求量;自变量包括 X_1(每日就诊人数)、X_2(月平均 X-射线检查次数)、X_3(月平均住院人数)和 X_4(地区人口)。以往研究显示,当模型中包含两个协变量 X_2 和 X_3 时,决定系数 $R^2=0.829$;当模型增加 X_1 和 X_4 两个协变量时,决定系数 $R^2=0.901$。设定检验水准 $\alpha=0.05$;检验效能取 $1-\beta=85\%$,试估计所需样本量。

nQuery Advisor7.0 实现:设定检验水准 $\alpha=0.05$;检验效能取 $1-\beta=85\%$。依题意知,原回归模型含 X_2 和 X_3 两个协变量,即 $A=2$,此时的决定系数为 $R_A^2=0.829$;之后在模型中增加 X_1 和 X_4 两个协变量,即$B=2$,此时模型的决定系数为 $R^2=0.901$,增加协变量后决定系数的增量为 $R_{AB}^2-R_A^2=0.072$。

在 nQuery Advisor7.0 主菜单选择:

Goal:Make Conclusion Using:⊙regression

Number of Groups:⊙One

Analysis Method:⊙Test

方法框中选择:linear regression, multiple covariates adjusted for others

在弹出的样本量计算窗口将各参数值键入,如图 5-9-1 所示,结果为 $n=19$,即本研究的样本量为 19 例。情形 2,对于本例,若检验 4 个自变量所建立的回归模型其决定系数是否为 0,这种情形即为多个自变量的线性回归,样本量计算方法如上一小节 5.1.1.4,已知 $R_{AB}^2=0.901$,设定检验水准 $\alpha=0.05$;检验效能取$1-\beta=85\%$,如图 5-9-2 所示,则所需样本量为 8 例。

Multiple regression, test 0 increase in R² for B covariates adjusting for A covariates	1	2	3	4	5
Test significance level, α	0.050				
Number prior covariates, A	2				
Correlation, R_A^2, for A covariates	0.8290				
Number of covariates to add, B	2				
Increase in $R^2 = R_{AB}^2 - R_A^2$	0.0720				
Power (%)	85				
n	19				

图 5-9-1　nQuery Advisor7.0 关于例 5-5 样本量估计的参数设置与计算结果

Multiple linear regression, test that R² = 0 for k normally distributed covariates	1	2	3	4	5
Test significance level, α	0.050				
Number of variables, k	4				
Squared multiple correlation, R²	0.9010				
Power (%)	85				
n	8				

图 5-9-2　nQuery Advisor7.0 关于例 5-5 情形 2 样本量估计的参数设置与计算结果

SAS9.2 软件实现:

```
%macro ROT4(a,ka,rqa,kb,rqb,power);
data ROT4;
a = &a;ka = &ka;rqa = &rqa;kb = &kb;rqb = &rqb;
power = &power;rqab = rqa + rqb;
if(a>1 | a<0) then error "Test significance Level must be in 0-1";
if(ka<1) then error "the number must be >= 1";
if(kb<1) then error "the number must be >= 1";
if(rqa<=0 or rqa>0.99) then error "the value must be in 0 to 0.99";
if(rqb<=0 or rqb>0.99) then error "the value must be in 0 to 0.99";
if(power>100 | power<1) then error "Power(%) must be in 1-100";
if(_error_=1) then stop;
n = ka + kb + 1;
do until(pw >= power/100);
n = n + 1;ncp = n * (rqab - rqa)/(1 - rqab);df1 =
```

△通信作者:陈平雁

```
kb; df2 = n - ka - kb - 1;
    pw = 1 - probf(finv(1 - a, df1, df2), df1, df2, ncp);
    end;
    proc print data = ROT4 noobs label; var a ka rqa kb rqb power n;
    label a = 'Test significance level,α'
    ka = 'Number prior covariates, A'
    rqa = 'Correlation, RA square, for A covariates'
    kb = 'Number of covariates to add, B'
    rqb = 'Increase in RB square = RAB square - RA square'
    power = 'Power(%)'
    n = 'n';
    quit;
    %mend ROT4;
    %ROT4(0.05,2,0.829,2,0.072,85);
```

SAS 运行结果：

Test significance level, α	Number prior covariates, A	Correlation, RA square, for A covariates	Number of covariates to add, B	Increase in RB square=RAB square-RA square	Power(%)	n
0.05	2	0.829	2	0.072	85	19

图 5－10　SAS9.2 关于例 5－5 样本量估计的参数设置与计算结果

5.1.1.6　用于检验回归系数的线性回归

方法：Dupont 和 Plummer（1998）[3] 给出的检验回归系数的样本量估计建立在自由度为 $n-2$，非中心参数为 $\sqrt{n}\delta$ 的非中心 t 分布基础上，其检验效能的计算公式为：

$$1-\beta = 1 - Probt(t_{1-\alpha/s,n-2}, n-2, \sqrt{n}\delta \qquad (5-9)$$

其中，

$$\delta = \frac{|b_1 - b_0|\sigma_x}{\sigma_{y.x}} \qquad (5-10)$$

式中，b_1 为备择假设的回归系数，b_0 为原假设的回归系数，σ_x 为自变量 X 的标准差，$\sigma_{y.x}$ 为残差标准误，δ 为效应量。

先设定样本量初始值，然后迭代样本量直到所得的检验效能满足条件为止，即为研究所需的样本量。

〔例 5－6〕为验证一种新的检测胆固醇水平的方法与金标准是否一致，采用线性回归方法，检验回归系数（即斜率）是否为 1。根据以往研究数据，该方法与金标准之间的相关系数是 0.9，该研究群体的胆固醇水平的标准误是 40 mg/dL。若设定检验水准 $\alpha = 0.05$，检验效能取 $1-\beta = 85\%$，双侧检验，试问如果回归系数大于 1.1 的话，需要多大样本量才足以检验出它与 1 的统计差异？

nQuery Advisor7.0 实现：设定检验水准 $\alpha = 0.05$；检验效能取 $1-\beta = 85\%$；采用双侧检验，即 $s = 2$。依据上述基础数据可知，$b_0 = 1$，$b_1 = 1.1$，$\sigma_x = 40$，$\rho = 0.9$。

在 nQuery Advisor7.0 主菜单选择：

Goal：Make Conclusion Using：⊙regression

Number of Groups：⊙One

Analysis Method：⊙Test

方法框中选择：linear regression，test for coefficient

为了估计残差的标准误，使用菜单栏中的 Assistants

Assistants：⊙Estimate SD

Estimate Standard Deviation：⊙ of residuals (errors)

在弹出的残差的标准误计算窗口将各参数值键入，如图 5－11 所示，结果为 $\sigma_{y.x} = 19.373$。

Residual (error) standard deviation around the line for one predictor x

	1	2	3	4
Regression coefficient, β	1.000			
Correlation coefficient, ρ	0.900			
Standard deviation of x, σ(x)	40.000			
Standard deviation of y, σ(y)	44.444			
SD of residuals (errors), σ	19.373			

图 5－11　nQuery Advisor7.0 关于例 5－6 残差的标准误计算结果

在弹出的样本量计算窗口将各参数值键入，如图 5－12 所示，结果为 $n = 213$，即本研究样本量为 213 例。

Linear regression test that β=β₀ for one x

	1	2	3
Test significance level, α	0.050		
1 or 2 sided test?	2		
Null hypothesis coefficient, β_0	1.000		
Alternative hypothesis coefficient, β	1.100		
Standard deviation of x's, σ(x)	40.000		
SD of residuals (errors), σ	19.373		
Effect size, δ	0.206		
Power (%)	85		
n	213		

图 5－12　nQuery Advisor7.0 关于例 5－6 检验效能估计的参数设置与计算结果

SAS9.2 软件实现：

```
%macro ROT5(a, s, b0, b1, sdx, sdy, r, yse, power); * 未知的输入0;
data ROT5;
a = &a; s = &s; b0 = &b0; b1 = &b1; sdx = &sdx; sdy = &sdy; r = &r; power = &power; yse = &yse;
```

```
if (yse =0) then do;
if (sdx =0) then sdx = r * sdy/b0;
if (sdy =0) then sdy = b0 * sdx/r;
if (r =0) then r = b0 * sdx/sdy;
yse = sqrt(sdy * *2 - r * sdx * sdy);
end;
theta = abs(b1 - b0) * sdx/yse;
if(a >1 | a<0) then error "Test significance Level must be in 0 - 1";
if(s^ =1 & s^ =2) then error "s =1 or 2";
if(yse < =0 ) then error "standard error must be >0";
if(sdx < =0 ) then error"standard deviation must be >0";
if(_error_ =1) then stop;
n =2;
do until(pw > =power/100);
n =n +1;ncp = sqrt(n) * theta;dt =n -2;
t1 =probt(tinv(a/s,dt),dt,ncp);t2 = probt(tinv(1 - a/s,dt),dt,ncp);
if (s =1) then pw =max(t1,(1 -t2));
else pw =t1 +1 - t2;
end;
proc print data =ROT5 noobs label;
var a s b0 b1 sdx yse theta power n;
label a = 'Test dignificance level,α'
s = '1 or 2 side test'
b0 = 'Null hypothesis coefficient,B0'
b1 = 'Alternative hypothesis coefficient,B'
sdx = "Standard deviation of x's,σ(x)"
yse = 'SD of residuals(errors),σ'
theta = 'Effect size'
pw = 'Power(%)'
n = 'n';
quit;
%mend ROT5;
%ROT5(0.05,2,1,1.1,40,0,0.9,0,85);
```

SAS 运行结果:

Test dignificance level,α	1 or 2 side test	Null hypothesis coefficient,B0	Alternative hypothesis coefficient,B	Standard deviation of x's,σ(x)	SD of residuals (errors),σ	Effect size	power	n
0.05	2	1	1.1	40	19.3729	0.20647	85	213

图5-13 SAS9.2 关于例5-6检验效能估计的参数设置与计算结果

5.2 两组的回归分析

5.2.1 差异性检验

5.2.1.1 单个自变量进行回归系数比较的线性回归

方法:Dupont 和 Plummer(1998)[3]给出的基于单个自变量进行回归系数比较的样本量估计建立在自由度为 $2n-4$,非中心参数为$\sqrt{\frac{n}{2}}\delta$的非中心 t 分布基础上,其检验效能的计算公式为:

$$1-\beta = 1-\mathrm{Pr}obt(t_{1-\alpha/s,2n-4},2n-4,\sqrt{\frac{n}{2}}\delta) \quad (5-11)$$

其中,

$$\delta = \frac{|b_1-b_2|\sigma_x}{\sigma_{y.x}} \quad (5-12)$$

式中,b_1 为第一组的回归系数,b_2 为第二组的回归系数,σ_x 为自变量 X 的标准差,$\sigma_{y.x}$为残差标准误,δ 为效应量。

先设定样本量初始值,然后迭代样本量直到所得的检验效能满足条件为止,即为研究所需的样本量。

〔例5-7〕某研究欲评估药物 A 与药物 B 的剂量-反应关系是否有差异。根据以往研究数据,当药物 B 的剂量分别是 2、4、8、16、32、64 时,剂量 X 经常用对数转化后建立的剂量-反应回归方程的回归系数为6.9,残差均方为25。如果设定 α 为0.05,检验效能取 $1-\beta=85\%$,双侧检验,并假定药物 A 剂量-反应回归方程的回归系数为6.9的2倍,即13.8,若要检验出两药物的回归系数有统计差异,所需样本量是多少?

nQuery Advisor7.0 实现:设定检验水准 $\alpha=0.05$;检验效能取 $1-\beta=85\%$;采用双侧检验,即 $s=2$。依据题意可知,$b_1=6.9$,$b_2=13.8$,$\sigma_{y.x}=5$。

在 nQuery Advisor7.0 主菜单选择:

Goal:Make Conclusion Using:⊙regression

Number of Groups:⊙two

Analysis Method:⊙Test

方法框中选择:Test for equality of slopes for one x

为了估计 X 的标准差,使用菜单栏中的 Assistants,

Assistants:⊙Estimate SD

Estimate Standard Deviation:⊙For specified x values

在弹出的 X 的标准差计算窗口将 X 的6个剂量输入第一栏,选择 log_10(X)并 add(添加),得到第二栏 X 的对数值,如图5-14所示,结果为 $\sigma_x=0.51$。

在样本量计算窗口的第1栏中将各参数值键入,

如图 5-15 所示，结果为 $n=38$，即本研究的每组样本量为 38 例。

Case	X0	Log X0		
1	2.00	0.30		
2	4.00	0.60		
3	8.00	0.90		
4	16.00	1.20		
5	32.00	1.51		
6	64.00	1.81		
7				
8				
N	6.00	6.00		
Mean	21.00	1.05		
Std.Dev.	23.72	0.56		
σ(x)	21.66	0.51		

图 5-14　nQuery Advisor7.0 关于例 5-7 中 X 的常用对数值的标准差

Linear regression test that $\beta_1=\beta_2$ for one x

	1	2	3
Test significance level, α	0.050		
1 or 2 sided test?	2		
Group 1 coefficient, β_1	6.900		
Group 2 coefficient, β_2	13.800		
Standard deviation of x's, σ(x)	0.510		
SD of residuals (errors), σ	5.000		
Effect size, δ	0.704		
Power (%)	85		
n per group	38		

图 5-15　nQuery Advisor7.0 关于例 5-7 检验效能估计的参数设置与计算结果

SAS9.2 软件实现：

```
%macro RTT0(a,s,b1,b2,sdx,yse,power);
data RTT0;
a = &a; s = &s; b1 = &b1; b2 = &b2; sdx = &sdx;
yse = &yse; power = &power;
theta = abs(b1 - b2) * sdx/yse;
if(a > 1 | a < 0) then error "Test significance Level must be in 0 - 1";
if(s^ = 1 & s^ = 2) then error "s = 1 or 2";
if(yse < = 0 ) then error "standard error must be > 0";
if(sdx < = 0) then error "standard deviation must be > 0";
if(_error_ = 1) then stop;
n = 2;
do until(pw > = power/100);
n = n + 1; ncp = sqrt(n/2) * theta; dt = 2 * n - 4;
t1 = probt(tinv(a/s,dt),dt,ncp); t2 = probt(tinv(1 - a/s,dt),dt,ncp);
if(s = 1) then pw = max(t1,(1 - t2));
else pw = t1 + 1 - t2;
end;
proc print data = RTT0 noobs label;
var a s b1 b2 sdx yse theta power n;
label a = 'Test dignificance level,α'
s = '1 or 2 side test'
b1 = 'Group 1 coefficient,B1'
b2 = 'Group 2 coefficient,B2'
sdx = "Standard deviation of x's,σ(x)"
yse = 'SD of residuals(errors),σ'
theta = 'Effect size'
pw = 'Power(%)'
n = 'n';
quit;
%mend RTT0;
%RTT0(0.05,2,6.9,13.8,0.51,5,85);
```

SAS 运行结果：

Test dignificance level, α	1 or 2 side test	Group 1 coefficient, B1	Group 2 coefficient, B2	Standard deviation of x's, σ(x)	SD of residuals (errors), σ	Effect size	power	n
0.05	2	6.9	13.8	0.51	5	0.7038	85	38

图 5-16　SAS9.2 关于例 5-7 检验效能估计的参数设置与计算结果

参　考　文　献

1. Hsieh FY. Sample size tables for logistic regression. Statistics in Medicine, 1989, 8: 795-802.
2. Cohen J. Statistical power analysis for the behavioral sciences, 2nd edition. Lawrence Erlbaum Associates, Hillsdale, New Jersey, 1988.
3. Dupont W, Plummer W. Power and sample size calculations for studies involving linear regression. Controlled Clinical Trials, 1998, 19: 589-601.

（责任编辑：郭海强）